
1 背景
当使用单片机的应用需要执行一个长延时时，设计者可以选择多种解决方法。

一个解决方法是使用硬件定时器。因为典型定时器只有16位或24位，这个方法需要系统时钟足够低来产生长延时。例如，一个运行在1 MHz系统时钟的24位定时器可以产生数十秒的延时。这对于很多应用来说还不够，此外这个方法对于系统性能有影响。

另外一个方法使用软件计数一定数目的定时器溢出。但是它使软件变得复杂，同时在等待长延时时不能进入低功耗模式。

第三种方法是使用一个很低频率的外部振荡器，定时器设置成外部触发的计数器，这可以产生很长的延时。但是这也增加了系统的成本，因为还需要使用其他的器件。

这里介绍的方法显示了怎样用AVR AT90系列单片机（AT90S2313、AT90S4414和AT90S8515）产生和处理长延时。只使用片内定时器而无需软件处理，这样在延时期间单片机内核可以进入低功耗模式。因为定时器使用系统时钟来计时，所以无需其他外部器件。

因为有长延时的能力，例如，运行在20MIPS的AVR单片机可以产生长达半小时的延时。

2 应用

下面列出了一些需要使用到长延时的应用：

· 人机接口的超时

· 环境测量（噪声等级、污染）

· 管理和过程控制

3 AVR单片机的定时器/计数器

下面小节简要的说明了AVR单片机定时器的用法。更多详细资料请参考相关的单片机数据手册。

3.1 定时器/计数器

AT90系列单片机提供了两个通用定时器／计数器，一个8位和一个16位。每个定时器／计数器有独立的10位预分频器，可以用做基于内部时钟的定时器或基于外部触发的计数器。

3.2 定时器/计数器预分频

图1显示了定时器／计数器的预分频器，可以选择4个不同的预分频：CK/8、CK/64、CK/256、CK/1024，这里CK是振荡器时钟。对于两个定时器／计数器，可以选择CK或者外部信号作为时钟源。

图1. 定时器／计数器预分频器

3.3 8位定时器/计数器0

图2显示了定时器／计数器0的结构图。

8位定时器／计数器0可以选择CK、预分频CK和外部信号作为时钟源，此外也可以通过设置TCCR0寄存器停止它。溢出状态标志位在TIFR（Time/Counter Interrupt Flag Register）寄存器中，控制在TCCR0（Time/Counter0 Control Register）寄存器中，而中断允许／禁止在TIMSK（Timer/Counter Interrupt Mask Register）寄存器。

当定时器／计数器使用外部时钟时，外部信号要和CPU的时钟频率同步。为了保证对外部时钟正确采样，两个外部时钟信号之间必须大于1个CPU时钟周期。外部时钟信号在CPU的内部时钟上升沿采样。

8位定时器／计数器0在使用低预分频时同时有高精度和高速特性。同样，在高预分频时有低速和准确定时功能。

图2. 定时器／计数器0的结构图

3.4 16位定时器/计数器1

图3显示了定时器/计数器1的结构框图。

16位的定时器/计数器1可以从CK、CK预分频和外部获取时钟源。此外，它还可以通过设置寄存器（TCCR1A和TCCR1B）来停止。不同的标志位（溢出、比较匹配和捕捉）和控制信号也在控制寄存器（TCCR1A和TCCR1B）中设置，中断控制在寄存器TIMSK。

当定时器/计数器1使用外部时钟，外部信号要和CPU的时钟频率同步。这和定时器0相同（参考3.3小节）。

16位的定时器/计数器1在低分频时有高精度和高速特性。同样，在高预分频时有低速和准确定时功能。定时器/计数器1还支持输出比较功能。

图3. 定时器/计数器1结构图

3.5 硬件设置

为了从快速时钟获取长延时，系统的预分频器和两个定时器需要比例化。硬件的配置如图4。
定时器／计数器1配置为一个定时器，它被设置为由系统时钟的预分频驱动。在计数达到输出比较寄存器OCR1A后翻转OC1A引脚的输出。这时，由硬件重载定时器／计数器1的寄存器TCNT1为0并重新开始计数。

OC1A引脚的输出又连接到T0引脚的输入，OC1A引脚的翻转可以触发定时器／计数器0。定时器／计数器0被配置为一个计数器，它在T0的每个上升沿计数。当定时器／计数器0溢出后，TIFR寄存器就会被置位，并触发中断，这可以用来指示延时的时间。
最长的延时可以用下面公式计算：

T = 2 / Fs * T1P * OCR1A * （256 – TCNT0）

这里：

Fs  = 系统时钟

T1P = TCCR1B定义的预分频

定时器1最大预分频比是：1024

定时器1最大计数：65536

OC1A引脚的翻转会额外乘以2

定时器0最大计数：256

图4. AVR单片机长延时的硬件设置

3.6 例程

这个简短的程序显示了怎样配置定时器在1MHz系统时钟下实现1小时的延时（处理性能和80C51相当）。
在开始阶段，设置定时器/计数器和中断，然后进入空闲模式。在1小时后，达到延时的时间并触发中断，中断将唤醒单片机执行用户任务。在例程中将翻转PA.0端口。当任务完成后，单片机将再次进入空闲模式，这个过程将不断持续下去。下面是带有注释的源代码：
; Constants definitions ----------------------------------------

.EQU PORTA = $1B

.EQU DDRA = $1A

.EQU DDRD = $11

.EQU TCCR1A = $2F

.EQU OCR1AH = $2B

.EQU OCR1AL = $2A

.EQU MCUCR = $35

.EQU TIMSK = $39

.EQU TCCR0 = $33

.EQU TCCR1B = $2E

.EQU TCNT1L = $2C

.EQU TCNT1H = $2D

.EQU TCNT0 = $32

.EQU SREG = $3F

.EQU SPH = $3E

.EQU SPL = $3D

; Interrupt service routines -----------------------------------

.ORG $0000

rjmp start

reti ; INT0 service

reti ; INT1 service

reti ; T/C1 capture service

reti ; T/C1 compare match A service

reti ; T/C1 compare match B service

reti ; T/C1 overflow service

reti ; T/C0 overflow service

reti ; SPI transfer complete service

reti ; UART receive service

reti ; UART data reg empty service

reti ; UART transmit service

reti ; Analog comparator service

; Peripherals configuration ----------------------------------------

start:

ldi r17, $01 ; Register init

ldi r16, $01 ; Program PORTA.0 as an output

out DDRA, r16

ldi r16, $01 ; Initialize stack pointer ...

out SPH, r16 ; ... to 0x100

ldi r16, $00

out SPL, r16

ldi r16, $20 ; Program OC1A as an output

out DDRD, r16

ldi r16, $40 ; Program TCCR1A to toggle ...

out TCCR1A, r16 ; ... OC1A on each compare match

ldi r16, $6D ; Program the output compare ...

out OCR1AH, r16 ; ... register for a division ...

ldi r16, $DD ; ... ratio of 28125

out OCR1AL, r16

ldi r16, $20 ; Configure sleep mode

out MCUCR, r16

ldi r16, $02 ; Enable T/C0 interrupt

out TIMSK, r16

ldi r16, $80 ; Global interrupt enable

out SREG, r16

; Infinite loop ----------------------------------------------------

loop:

rcall main ; Call the main routine

ldi r16, $06 ; Reload counter 0 for a division ...

out TCNT0, r16 ; ... ratio of 250

ldi r16, $06 ; Start counter 0 for ...

out TCCR0, r16 ; ... external pin T0 source

ldi r16, $00 ; Reset timer 1 value

out TCNT1H, r16

ldi r16, $00

out TCNT1L, r16

ldi R16, $0C ; Start timer 1 for a prescale ...

out TCCR1B, r16 ; ... ratio of 256

sleep ; Wait for delay

ldi r16, $00 ; Stop timer 1

out TCCR1B, r16

ldi r16, $00 ; Stop timer 0

out TCCR0, r16

rjmp loop

; Main routine -------------------------------------------------

; This routine just toggles PORTA.0

main:

in r16, PORTA

eor r16, r17

out PORTA, r16

ret
以下是使用GCC的代码，可以和汇编的代码做比较（由译者提供）：
//CPU: Atmega8515  OSC: 1MHz

#include <avr/io.h>

#include <avr/interrupt.h>

#include <avr/sleep.h>

#ifndef F_CPU

#define F_CPU 1000000UL

#endif

int main(void)

{

  DDRA   = (1<<PA0);    //PA0做输出
  DDRD   = (1<<PD5);    //OC1A做输出
  TCCR1A = 0x40;        //比较输出模式，匹配时OC1A取反
  OCR1A = 0x6DDD;       //比较系数28125
                        //延时时间：2*256*28125*250/1000000=3600秒

  sleep_enable();       //允许睡眠模式
  set_sleep_mode(SLEEP_MODE_IDLE);

  TIMSK = (1<<TOIE0);   //允许T/C0溢出中断
  sei();                //全局中断使能
  while(1)

  {

    PORTA^= (1<<PA0);   //PA0取反
    TCNT0 = 0x06;       //计数250
    TCCR0 = (1<<CS02)|(1<<CS01); //启动计数器0,T0做输入
    TCNT1 = 0;          //定时器1重新计数
    TCCR1B= (1<<WGM12)|(1<<CS12); //256分频，CTC模式
    sleep_cpu();

    TCCR1B= 0;          //停止定时器1
    TCCR0 = 0;          //停止计数器0
  }

}
译者注：

本文主要的思路就是利用AVR单片机硬件的CTC功能，由定时器1的CTC输出作为定时器0的输入，产生长延时，中间无需任何软件开销。

（全文完）

